

AD-A037 754

AIR FORCE WEAPONS LAB KIRTLAND AFB N MEX
THERMAL RADIATIVE LOSSES FROM GRAPHITE CALORIMETERS USED IN PUL--ETC(U)
NOV 76 J A MACFARLANE
AFWL-TR-76-180

F/G 20/13
PUL--ETC(U)


UNCLASSIFIED

NL

| OF |
AD
A037754



END
DATE
FILMED
4-77



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

ADA 037754

AFWL-TR-76-180

AFWL-TR-76-180

12



THERMAL RADIATIVE LOSSES FROM GRAPHITE CALORIMETERS USED IN PULSED ELECTRON BEAM EXPERIMENTS

November 1976

Final Report

DDC
APR 5 1977
C

Approved for public release; distribution unlimited.

This research was sponsored by the Defense Nuclear Agency under Subtask N99QAXAA121, Work Unit 10, "Diagnostics Development."

COPY AVAILABLE TO DDC DOES NOT PERMIT FULLY LEGIBLE PRODUCTION

Prepared for
Director
DEFENSE NUCLEAR AGENCY
Washington, DC 20305

AIR FORCE WEAPONS LABORATORY
Air Force Systems Command
Kirtland Air Force Base, NM 87117


DDC FILE COPY

This final report was prepared by the Air Force Weapons Laboratory, Kirtland Air Force Base, New Mexico, under Job Order WDNA4501. Lt John A. MacFarlane (DYV) was the Laboratory Project Officer-in-Charge.

When US Government drawings, specifications, or other data are used for any purpose other than a definitely related Government procurement operation, the Government thereby incurs no responsibility nor any obligation whatsoever, and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

This report has been reviewed by the Information Office (OI) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.


This technical report has been reviewed and is approved for publication.




JOHN A. MACFARLANE
Lt, USAF
Project Officer

ACCESSION FOR	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>
NTIS	Public Section		
DDC	Diff Section		
UNCLASSIFIED			
JUSTIFICATION			
BY	DISTRIBUTION/AVAILABILITY CODES		
	Dist. AVAIL. CODE/SP. SPECIAL		
	A		

FOR THE COMMANDER



CARL D. STUBER
Major, USAF
Chief, Materials Vulnerability/
Hardening Branch



JOHN S. DEWITT
Lt Colonel, USAF
Chief, Technology Division

DO NOT RETURN THIS COPY. RETAIN OR DESTROY.



UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER 14 AFWL-TR-76-180	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER 9
4. TITLE (and Subtitle) THERMAL RADIATIVE LOSSES FROM GRAPHITE CALORIMETERS USED IN PULSED ELECTRON BEAM EXPERIMENTS.		5. TYPE OF REPORT & PERIOD COVERED Final Report
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) 10 John A. MacFarlane, Lt, USAF		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS AIR FORCE WEAPONS LABORATORY Kirtland AFB, NM 87117		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 62704H/WDNA4501/Subtask N99QAXAA121, Work Unit 10
11. CONTROLLING OFFICE NAME AND ADDRESS		12. REPORT DATE 91 November 1976
		13. NUMBER OF PAGES 50 (93) 47p.
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office) Director Defense Nuclear Agency Washington, D.C. 20305		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report) Approved for public release; distribution unlimited.		
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)		
18. SUPPLEMENTARY NOTES This research was sponsored by the Defense Nuclear Agency under Subtask N99QAXAA121, Work Unit 10, "Diagnostics Development." 013 150		
19. KEY WORDS (Continue on reverse side if necessary and identify by block number) Heat Transfer Diagnostics Cooling Curves Calorimeter Block Electron Beams Finite Difference Equations Heat Loss Computer Model		
20. ABSTRACT (Continue on reverse side if necessary and identify by block number) A study to analytically model the thermal behavior of a graphite calorimeter block subjected to an initially non-uniform temperature profile has been conducted. The model allows prediction of the amount of energy lost from the block by the time a uniform temperature profile is achieved. It is assumed that the only means of energy loss to the surroundings is that of thermal radiation; the effects of convection and thermal conduction to the surroundings are neglected. In addition, it is assumed that the calorimeter is exposed to		

over
602

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

BLOCK 20. ABSTRACT (cont)

a uniform beam of energy with an area equal to that of the calorimeter face. Thus, conduction in the lateral directions is neglected. Finally, it is assumed that the beam energy is deposited instantaneously, with the resultant temperature profile being taken as an initial condition. The results of a parametric study in which the radiation losses are quantified for a variety of electron beam environments are presented. Finally, an evaluation of the manner in which calorimetry data are presently reduced is made.

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

CONTENTS

<u>Section</u>	<u>Page</u>
I. INTRODUCTION	1
II. DEVELOPMENT OF MATHEMATICAL MODEL	5
1. Introduction	5
2. Development of Finite Difference Equations	5
a. Internal Elements	6
b. Elements at Front and Rear Free Surfaces	12
3. Solution of Finite Difference Equations	15
4. Summary	18
III. MATERIAL PROPERTIES OF GRAPHITE	19
IV. COMPUTER MODEL	23
V. RESULTS	28
VI. CONCLUSIONS AND RECOMMENDATIONS	39
VII. LIST OF SYMBOLS	41

ILLUSTRATIONS

<u>Figure</u>		<u>Page</u>
1	Simplified Model of Calorimeter	2
2	Typical Temperature History Given By Thermocouple Attached to Rear Surface of Calorimeter.	2
3	Differential Element Model of Calorimeter	7
4	Energy Balance on ith Internal Element	8
5	Energy Balance on First Element	8
6	Specific Heat and Thermal Conductivity of Graphite Versus Temperature	20
7	Electron Beam Deposition Profiles Used in This Study	29
8	Temperature Profiles for Calorimeter Block Subjected to a "1 MeV", 1200 cal/g Electron Beam	30
9	Back Surface Temperature Histories for a "500 keV" Electron Beam	32
10	Percent of Initially Deposited Energy Lost by Radiation at Time of Maximum Back Surface Temperature	34
11	Back Surface Temperature History for Calorimeter Block Subjected to "1 MeV", 1800 cal/g Electron Beam	36
12	Effect of Calorimeter Area Upon Percent Energy Lost at Time of Maximum Back Surface Temperature	37

SECTION I

INTRODUCTION

High intensity, pulsed electron beams are widely used to study the dynamic response of materials and structures subjected to rapid, in-depth heating. When performed effectively, such experiments can be of considerable value in studies relating to the nuclear hardness assessments of reentry vehicle systems.

The manner in which the beam energy is distributed throughout the depth of a sample target is dependent upon the electron beam current and the voltage which accelerates the electrons. In general, the energy deposition is nonuniform, decreasing with depth into the sample. The time required for this deposition to occur is variable, but is nominally 50-200 nanoseconds.

For any electron beam experiment to produce meaningful results, it is necessary to know the thermal loading conditions of the sample. These loading conditions have traditionally been determined through calorimetric measurements made on separate beam "diagnostic" shots. On such a shot, a calorimeter block, or an array of such blocks, is used to measure the total energy in the beam at the sample location.

A typical calorimeter block is shown in Figure 1. The block is generally constructed of graphite because of the high vaporization energy of that material. The energy deposited in the calorimeter is deduced from a measurement of the block temperature. This temperature is generally measured by means of a thermocouple attached to the rear surface of the block, or placed inside of the block by means of a hollow screw.




Figure 1. Simplified Model of Calorimeter




Figure 2. Typical Temperature History Given by Thermocouple Attached to Rear Surface of Calorimeter

Shown in Figure 2 is a typical temperature history as given by a thermocouple attached to the rear surface of a calorimeter block. Because the beam energy is initially deposited in a nonuniform manner, some time must elapse before it is uniformly distributed throughout the block by means of thermal conduction. As shown in Figure 2, the back surface temperature rapidly rises with time until the energy is uniformly distributed, and then slowly decreases as energy is lost from the calorimeter block to the surroundings.

Because the thermocouple measures the temperature at a given point in the depth of the block, there is considerable uncertainty involved in detecting the total energy deposited unless the block is at a uniform temperature. For the case presented in Figure 2, it is generally assumed that the block is at a uniform temperature when the maximum back surface temperature is achieved. The total energy in the block at that time is then obtained using the specific heat of graphite at that temperature. To obtain the energy initially deposited in the calorimeter block, various extrapolations of the "cooling curve" are employed. The most common extrapolation, a linear one, is shown in Figure 2. This extrapolation assumes that the block cools at a constant rate and that the initial energy in the block can be given by the initial temperature obtained from such an extrapolation.

It is the purpose of this study to analytically model the thermal behavior of a graphite calorimeter block subjected to an initially non-uniform temperature profile. This model will allow prediction of the amount

of energy lost from the block by the time a uniform temperature profile is achieved. In addition, it will determine the validity of the assumption that the block is at a uniform temperature when the maximum back surface temperature is reached. Then, through parametric variations, the model will be used to determine in what electron beam environments graphite calorimetry can be used with confidence.

For this study, the simplified calorimeter model of Figure 1 is employed. It is assumed that the only means of energy loss to the surroundings is that of thermal radiation; the effects of convection and thermal conduction to the surroundings are neglected. In addition, it is assumed that the calorimeter is exposed to a uniform beam with an area equal to that of the calorimeter face. Thus, conduction in the lateral directions is neglected. Finally, it is assumed that the beam energy is deposited instantaneously, with the resultant temperature profile being taken as an initial condition.

In Section II, a mathematical model is developed for a calorimeter block subjected to these conditions and constraints. In Section III, the model is applied to a graphite block through specification of the thermal properties of that material. Section IV describes the manner in which a computer model is developed to determine the effectiveness of graphite calorimetry. Section V presents the results of a parametric study in which the radiation losses are quantified for a variety of electron beam environments. In addition, Section V presents an evaluation of the manner in which calorimetry data are presently reduced to yield information about the loading environment. Finally, Section VI presents conclusions and recommendations arising from this study.

SECTION II

DEVELOPMENT OF MATHEMATICAL MODEL

1. INTRODUCTION

For the problem of one-dimensional heat conduction into a graphite calorimeter block, proper account should be taken of the variation of thermal conductivity and specific heat with temperature. For this reason, the one-dimensional heat conduction equation cannot be reduced to the common Laplace equation. Rather, it must be written in its most general form:

$$\frac{\partial T}{\partial t} = \frac{1}{\rho C} \frac{\partial}{\partial x} \left(k \frac{\partial T}{\partial x} \right) \quad (1)$$

For the problem being addressed, the initial and boundary conditions imposed upon equation (1) prevent it from being solved through analytical methods. The radiation heat transfer at the free surfaces provides a nonlinear boundary condition, while the initial temperature profile is generally nonuniform. For these reasons, the problem being considered is best analyzed through the use of finite difference techniques.

2. DEVELOPMENT OF FINITE DIFFERENCE EQUATIONS

Rather than apply finite difference approximations to equation (1), these approximations should be applied to the First Law of Thermodynamics for the calorimeter block. This approach allows consideration of radiative losses from the sides of the block without resorting to a two-dimensional formulation.

Consider the calorimeter block shown in Figure 3. If this block is divided into M differential elements, each Δx thick, an energy balance can be developed for each element. By choosing Δx to be small, the i th element can be considered to be at a uniform temperature T_i . Because of geometrical similarity, the development will be identical for each internal element. To account for the front and rear free surfaces, however, a separate development is necessary for the 1st and Mth elements.

a. Internal Elements

Figure 4 illustrates the various mechanisms by which energy is transferred into and out of the i th internal element. Since this element is at a uniform temperature T_i , conduction terms for the y and z directions can be neglected. The conduction into the left face of the i th element, given by $Q_{i(x)}^k$, results from a temperature difference between element $i-1$ and element i . Similarly, $Q_{i(x+\Delta x)}^k$ represents conduction out of the right face of the i th element. The energy loss to the surroundings, assumed to occur through thermal radiation only, is given by Q_i^r .

The First Law of Thermodynamics, as written for the i th internal element, is

$$\frac{\partial E_i}{\partial t} = Q_{i(x)}^k - Q_{i(x+\Delta x)}^k - Q_i^r \quad (2)$$

With i denoting position and n denoting time, the left hand side of equation (2) is approximated with the following fully implicit finite difference scheme:




Figure 3. Differential Element Model of Calorimeter




Figure 4. Energy Balance on i th Internal Element




Figure 5. Energy Balance on First Element

$$\frac{\partial E_i}{\partial t} = \rho C_i \Delta x YZ \frac{\partial T_i}{\partial t} \approx \rho C_{i,n} \Delta x YZ \left[\frac{T_{i,n+1} - T_{i,n}}{\Delta t} \right] \quad (3)$$

Note that the subscript n denotes a known quantity, while the subscript $n+1$ denotes an unknown quantity at the next time step. Using the same difference scheme, the energy conduction terms are approximated by

$$Q_{i(x)}^k = YZ \left(\frac{-k \partial T}{\partial x} \right)_{(x)}$$

$$\approx YZ \left[\frac{k_{i,n} + k_{i-1,n}}{2} \right] \left[\frac{T_{i-1,n+1} - T_{i,n+1}}{\Delta x} \right] \quad (4)$$

$$Q_{i(x+\Delta x)}^k = YZ \left(-k \frac{\partial T}{\partial x} \right)_{(x+\Delta x)}$$

$$\approx YZ \left[\frac{k_{i,n} + k_{i+1,n}}{2} \right] \left[\frac{T_{i,n+1} - T_{i+1,n+1}}{\Delta x} \right] \quad (5)$$

For the sake of simplicity, without any loss of accuracy, the thermal conductivity of the i th element is calculated using the known temperature, $T_{i,n}$. The radiation loss from the sides of the element is given by

$$Q_i^r = \sigma \epsilon 2\Delta x (Y+Z) \left[T_{i,n+1}^4 - T_a^4 \right] \quad (6)$$

Putting equations (3) through (6) into equation (2) and rearranging yields the following implicit finite difference equation:

$$T_{i,n+1} - T_{i,n} =$$

$$\left(k_{i,n} + k_{i-1,n} \right) \left\{ \frac{\Delta t}{2\rho C_{i,n} \Delta x^2} \right\} T_{i-1,n+1}$$

$$- \left(k_{i+1,n} + 2k_{i,n} + k_{i-1,n} \right) \left\{ \frac{\Delta t}{2\rho C_{i,n} \Delta x^2} \right\} T_{i,n+1}$$

$$+ \left(k_{i,n} + k_{i+1,n} \right) \left\{ \frac{\Delta t}{2\rho C_{i,n} \Delta x^2} \right\} T_{i+1,n+1}$$

$$- \left\{ \frac{\Delta t \sigma \epsilon}{2\rho C_{i,n} \Delta x^2} \right\} \left(\frac{4\Delta x^2 (Y+Z)}{YZ} \right) \left[T_{i,n+1}^4 - T_a^4 \right] \quad (7)$$

Equation (7) is a nonlinear difference equation because of the radiation term involving $T_{i,n+1}^4$. By using the common radiation approximation [1],

$$T_{i,n+1}^4 = 4T_{i,n}^3 T_{i,n+1} - 3T_{i,n}^4 \quad (8)$$

equation (7) can be linearized*.

I. Gaumer, G. R., "Stability of Three Finite Difference Methods of Solving Transient Temperatures," ARS Journal, Vol. 32, No. 10, October 1962, pp 1595-1597.

*The radiation approximation is obtained by taking the time derivative of $T_{i,n}^4$.

Equations (7) and (8) can be combined in a considerably simplified form if the following terms are defined:

$$S_1 = \frac{4\Delta x^2(Y+Z)}{YZ} \quad (9)$$

$$S_{2i} = 4\sigma\varepsilon T_{i,n}^3 \quad (10)$$

$$S_{3i} = \sigma\varepsilon \left[-3T_{i,n}^4 - T_a^4 \right] \quad (11)$$

$$S_{4i} = \frac{\Delta t}{2\rho C_{i,n} \Delta x^2} \quad (12)$$

$$S_{5i} = k_{i,n} + k_{i+1,n} \quad (13)$$

$$S_{6i} = k_{i-1,n} + 2k_{i,n} + k_{i+1,n} \quad (14)$$

$$S_{7i} = k_{i,n} + k_{i-1,n} \quad (15)$$

Each of these quantities are known, being labeled with the subscript n. These definitions allow equations (7) and (8) to be combined in the following form:

$$\begin{aligned}
& T_{i-1,n+1} (S_{4i} \cdot S_{7i}) \\
& + T_{i,n+1} (-1 - S_{4i} \cdot S_{6i} - S_{4i} \cdot S_1 \cdot S_{2i}) \\
& + T_{i+1,n+1} (S_{4i} \cdot S_{5i}) \\
& = -T_{i,n} + (S_{4i} \cdot S_1 \cdot S_{3i}) \tag{16}
\end{aligned}$$

This equation is the final form of the finite difference approximation to the First Law of Thermodynamics for the i th element. It can be used to describe each of the internal elements depicted in Figure 3. To completely describe the calorimeter block, additional equations must be developed for the first and M th elements. These developments are presented in the following subsection.

b. Elements at Front and Rear Free Surfaces

Figure 5 illustrates the mechanisms by which energy is transferred into and out of the first element. This element, at a uniform temperature T_1 , experiences conduction through the right face only. Energy transfer by radiation occurs at the front surface as well as from the free side surfaces of the element.

The First Law of Thermodynamics, as written for the first element, is

$$\frac{\partial E_1}{\partial t} = -Q_{1(x=\Delta x)}^k - Q_1^r \quad (17)$$

Using the same finite difference approximations employed for the internal elements, the components of equation (17) are given by

$$\frac{\partial E_1}{\partial t} = \rho C_{1, \Delta x} YZ \left(\frac{\partial T_1}{\partial t} \right) \approx \rho C_{1, n} \Delta x YZ \left[\frac{T_{1, n+1} - T_{1, n}}{\Delta t} \right] \quad (18)$$

$$Q_{1(x=\Delta x)}^k = YZ \left(-k \frac{\partial T}{\partial x} \right)_{(x=\Delta x)} \approx YZ \left[\frac{k_{1, n} + k_{2, n}}{2} \right] \left[\frac{T_{1, n+1} - T_{2, n+1}}{\Delta x} \right] \quad (19)$$

$$Q_1^r = \sigma \epsilon 2 \Delta x (Y+Z) + YZ \left[T_{1, n+1}^4 - T_a^4 \right] \quad (20)$$

Putting equations (18) through (20) into equation (17) yields a relation which is similar to equation (7)

$$\begin{aligned} T_{1, n+1} - T_{1, n} = & \\ & - \left(k_{1, n} + k_{2, n} \right) \left\{ \frac{\Delta t}{2 \rho C_{1, n} \Delta x^2} \right\} T_{1, n+1} \\ & + \left(k_{1, n} + k_{2, n} \right) \left\{ \frac{\Delta t}{2 \rho C_{1, n} \Delta x^2} \right\} T_{2, n+1} \\ & - \left\{ \frac{\Delta t \sigma \epsilon}{2 \rho C_{1, n} \Delta x^2} \right\} \left(\frac{4 \Delta x^2 (Y+Z) + 2 YZ \Delta x}{YZ} \right) \left[T_{1, n+1}^4 - T_a^4 \right] \end{aligned} \quad (21)$$

After introducing the radiation approximation of equation (8), the following definitions are introduced to simplify equation (21):

$$S_{21} = 4\sigma\epsilon T_{1,n}^3 \quad (22)$$

$$S_{31} = \sigma\epsilon \left[-3T_{1,n}^4 - T_a^4 \right] \quad (23)$$

$$S_{41} = \frac{\Delta t}{2\rho C_{1,n} \Delta x^2} \quad (24)$$

$$S_{51} = k_{1,n} + k_{2,n} \quad (25)$$

Note the similarity between these definitions and those employed for the internal elements. Putting equations (22) through (25) into equation (21) yields the following final form of the finite difference equation for the first element:

$$\begin{aligned} & T_{1,n+1} (-1 - S_{41} \cdot S_{51} - S_{41} \cdot (S_1 + 2\Delta x) \cdot S_{21}) \\ & + T_{2,n+1} (S_{41} \cdot S_{51}) \\ & = -T_{1,n} + S_{41} \cdot (S_1 + 2\Delta x) \cdot S_{31} \end{aligned} \quad (26)$$

Using a development which parallels that for the first element, the difference equation for the last element ($i = M$) can be written as

$$\begin{aligned}
 & T_{M-1,n+1} (S_{4M} \cdot S_{7M}) \\
 & + T_{M,n+1} (-1 - S_{4M} \cdot S_{7M} - S_{4M} \cdot (S_1 + 2\Delta x) \cdot S_{2M}) \\
 & = -T_{M,n} + S_{4M} \cdot (S_1 + 2\Delta x) \cdot S_{3M}
 \end{aligned} \tag{27}$$

where

$$S_{2M} = 4\sigma\epsilon T_{M,n}^3 \tag{28}$$

$$S_{3M} = \sigma\epsilon \left[-3T_{M,n}^4 - T_a^4 \right] \tag{29}$$

$$S_{4M} = \frac{\Delta t}{2\rho C_{M,n} \Delta x^2} \tag{30}$$

$$S_{7M} = k_{M,n} + k_{M-1,n} \tag{31}$$

3. SOLUTION OF FINITE DIFFERENCE EQUATIONS

Equations (16), (26), and (27) constitute a set of M equations which can be used to describe the thermal behavior of the calorimeter block depicted in Figure 3. Note that each of the equations is of the form

$$A_i T_{i-1,n+1} + B_i T_{i,n+1} + C_i T_{i+1,n+1} = D_i \quad (32)$$

where

for $i = 1$:

$$A_1 = 0 \quad (33)$$

$$B_1 = -1 - S_{41} \cdot S_{51} - S_{41} \cdot (S_1 + 2\Delta x) \cdot S_{21} \quad (34)$$

$$C_1 = S_{41} \cdot S_{51} \quad (35)$$

$$D_1 = -T_{1,n} + S_{41} \cdot (S_1 + 2\Delta x) \cdot S_{31} \quad (36)$$

for $1 < i < M$:

$$A_i = S_{4i} \cdot S_{7i} \quad (37)$$

$$B_i = -1 - S_{4i} \cdot S_{6i} - S_{4i} \cdot S_1 \cdot S_{2i} \quad (38)$$

$$C_i = S_{4i} \cdot S_{5i} \quad (39)$$

$$D_i = -T_{i,n} + S_{4i} \cdot S_1 \cdot S_{3i} \quad (40)$$

for i = M:

$$A_M = S_{4M} \cdot S_{7M} \quad (41)$$

$$B_M = -1 - S_{4M} \cdot S_{7M} - S_{4M} \cdot (S_1 + 2\Delta x) \cdot S_{2M} \quad (42)$$

$$C_M = 0 \quad (43)$$

$$D_M = -T_{M,n} + S_{4M} \cdot (S_1 + 2\Delta x) \cdot S_{3M} \quad (44)$$

Because equations (16), (26), and (27) can be written in the form of equation (32), with coefficients described by equations (33) through (44), they constitute a tridiagonal matrix. Use of these equations to predict the temperature profile history for the calorimeter block requires that this matrix be solved at each new time step. The matrix is easily solved through use of the Thomas Algorithm [2].

For each new time step, solution for $T_{i,n+1}$ is performed as follows:

$$\beta_i = B_i - \frac{A_i C_{i-1}}{\beta_{i-1}}, \text{ where } \beta_1 = B_1 \quad (45)$$

$$\gamma_i = \frac{D_i - A_i \gamma_{i-1}}{\beta_i}, \text{ where } \gamma_1 = \frac{D_1}{B_1} \quad (46)$$

2. Von Rosenberg, D. U., Methods for the Numerical Solution of Partial Differential Equations, pp 113, American Elsevier Publishing Company, Inc., New York, 1969.

After computing β_i and γ_i for all i (called the First Half of the Thomas Algorithm), one then computes the new temperatures by

$$T_{M,n+1} = \gamma_M \quad (47)$$

$$T_{i,n+1} = \gamma_i - \frac{C_i T_{i+1,n+1}}{\beta_i} \quad (48)$$

(called the Back Half of the Thomas Algorithm).

4. SUMMARY

In this section, the First Law of Thermodynamics has been applied to each of the M difference elements represented in Figures (3) through (5). This has resulted in a mathematical model which is to be used to predict the thermal behavior of graphite calorimetry under a variety of conditions. Before such use can be made of this model, information is required regarding the material properties of graphite. This information is given in the following section.

SECTION III
MATERIAL PROPERTIES OF GRAPHITE

Before the mathematical model developed in Section II can be applied to the specific case of a graphite calorimeter, values must be assigned to certain material properties. These properties are the density, ρ , the emissivity, ϵ , the thermal conductivity, k , and the specific heat, c .

Graphite is produced in different forms through various means. For this reason, different values of density are quoted in the literature. The value used in this study is 1.8 g/cm^3 .

The total hemispherical emissivity for graphite is nominally 0.70 to 0.95. For this study, a value of 0.85 was chosen.

As mentioned in Section II, the thermal conductivity of graphite is a function of temperature. For the range of temperatures encountered by graphite calorimeters, the variation of the thermal conductivity can be quite significant. Reference 3 gives values of k versus T for various forms of graphite; for each case, k was measured in the direction of the grain axis. A bilinear relationship was found to well describe the behavior of k , which varies from $0.40 \text{ cal/cm sec } ^\circ\text{K}$ at 0°K to $0.05 \text{ cal/cm sec } ^\circ\text{K}$ at 4000°K . This relationship is shown in Figure 6 and is represented mathematically as

$$0^\circ\text{K} \leq T \leq 1000^\circ\text{K} \quad k = 0.400 - 2.50 \times 10^{-4} T \text{ (cal/cm sec } ^\circ\text{K)} \quad (49)$$

3. Goldsmith, A., Waterman, T. E., Hirschnorn, H. J., Handbook of Thermophysical Properties of Solid Materials, Vol I, pp 115-121, The MacMillan Company, New York, 1961.




Figure 6. Specific Heat and Thermal Conductivity of Graphite Versus Temperature

$$1000^{\circ}\text{K} \leq T \leq 4000^{\circ}\text{K} \quad k = 0.183 - 3.33 \times 10^{-5} T \quad (\text{cal/cm sec}^{\circ}\text{K}) \quad (50)$$

The specific heat of graphite is also a function of temperature, primarily in the range of $0^{\circ}\text{K} < T < 1000^{\circ}\text{K}$. Reference 4 presents a specific heat versus temperature relationship for graphite. Again, a bilinear relationship was found to well describe the thermal behavior. Shown in Figure 6, this relationship is

$$0^{\circ}\text{K} \leq T \leq 1000^{\circ}\text{K} \quad C = 5.0 \times 10^{-4} T \quad (\text{cal/g}^{\circ}\text{K}) \quad (51)$$

$$1000^{\circ}\text{K} \leq T \leq 4000^{\circ}\text{K} \quad C = 0.50 \quad (\text{cal/g}^{\circ}\text{K}) \quad (52)$$

Using the definition of specific heat, the internal energy of graphite can be expressed as a function of temperature also:

$$u(T) = \int_{T_0}^T C \, dT \quad (53)$$

where $u(T)$ is calculated relative to some arbitrary temperature T_0 .

Taking T_0 to be the ambient temperature, T_a , equations (51)-(53) yield the following relationships for $u(T)$:

$$0^{\circ}\text{K} \leq T \leq 1000^{\circ}\text{K} \quad u = 2.5 \times 10^{-4} (T^2 - T_a^2) \quad (\text{cal/g}) \quad (54)$$

4. Hultgren, R., Selected Values of the Thermodynamic Properties of the Elements, American Society for Metals, Metals Park, OH, p 91, 1973.

$$1000^{\circ}\text{K} \leq T \leq 4000^{\circ}\text{K} \quad u = 0.50T - 250(1.0 + 1.0 \times 10^{-6} T_a^2) (\text{cal/g}) \quad (55)$$

These equations can be written in terms of T versus u:

$$0 \text{ cal/g} \leq u \leq 228.5 \text{ cal/g} \quad T = (4.0 \times 10^3 u + T_a^2)^{1/2} (^{\circ}\text{K}) \quad (56)$$

$$228.5 \text{ cal/g} \leq u \quad T = 2u + 500 (1 + 1 \times 10^{-6} T_a^2) (^{\circ}\text{K}) \quad (57)$$

When these material property relationships are combined with the mathematical model of Section II, the result is a calculational tool for predicting the thermal behavior of a graphite calorimeter. How this calculational tool is incorporated into a computer model for performing such predictions is the subject of the following section.

SECTION IV
COMPUTER MODEL

As stated earlier, combining the mathematical model developed in Section II with the material property relationships defined in Section III results in a calculational tool for predicting the thermal behavior of a graphite calorimeter. This calculational tool has been incorporated into a computer model to allow for rapid solution of a large number of problems.

The computer model used for this study is presented in the following pages. The model is fairly straightforward and should be self-explanatory. For convenience, all symbols employed in the model are defined. In addition, various sections of the model have been identified with regard to their function.

The following section presents the results of utilizing this model for a variety of system dimensions and initial thermal loading conditions.

```

PROGRAM MAC3(INPUT,OUTPUT)
C
C*** IDENTIFICATION OF SYMBOLS USED IN COMPUTER CODE
C
C ** W - THICKNESS OF CALORIMETER BLOCK,CM
C ** Y - HEIGHT OF CALORIMETER BLOCK,CM
C ** Z - WIDTH OF CALORIMETER BLOCK,CM
C ** RHO - DENSITY OF CALORIMETER BLOCK MATERIAL,G/CM**3
C ** EM1 - TOTAL HEMISPHERICAL EMISSIVITY OF CALORIMETER SURFACE
C ** TA - AMBIENT TEMPERATURE, K
C ** SIG - STEFAN BOLTZMANN CONSTANT, CAL/CM**2SEC K**4
C ** I - SUBSCRIPT DENOTING POSITION IN CALORIMETER THICKNESS
C ** X(I) - LOCATION OF ITH ELEMENT, CM
C ** DX - THICKNESS OF EACH DIFFERENCE ELEMENT,CM
C ** M - TOTAL NUMBER OF DIFFERENCE ELEMENTS
C ** N - SUBSCRIPT DENOTING TIME
C ** TIME - TIME ELAPSED SINCE ENERGY DEPOSITION, SEC
C ** DT - DURATION OF TIME STEP (VARIABLE),SEC
C ** K - TOTAL NUMBER OF TIME STEPS ALLOWED
C ** AB - VARIABLE USED TO ADJUST NEXT TIME STEP
C ** ABT - VARIABLE USED TO ADJUST NEXT TIME STEP
C ** T1(I) - TEMPERATURE OF ITH ELEMENT AT OLD TIME (N)
C ** T2(I) - TEMPERATURE OF ITH ELEMENT AT NEW TIME (N+1)
C ** U1(I) - SPECIFIC ENERGY OF ITH ELEMENT AT OLD TIME (N),CAL/GM
C ** U2(I) - SPECIFIC ENERGY OF ITH ELEMENT AT NEW TIME (N+1) ,CAL/GM
C ** CP(I) - SPECIFIC HEAT OF ITH ELEMENT,CAL/G-K
C ** CK(I) - THERMAL CONDUCTIVITY OF ITH ELEMENT,CAL/CM**2-SEC-K
C ** UTOT1 - TOTAL ENERGY IN CALORIMETER AT OLD TIME,CAL
C ** UTOT2 - TOTAL ENERGY IN CALORIMETER AT NEW TIME,CAL
C ** ELOSS - PERCENT ENERGY LOST AT END OF GIVEN TIME STEP
C ** S1 - FINITE DIFFERENCE COEFFICIENT DEFINED IN TEXT
C ** S2(I) THROUGH S7(I) - FINITE DIFFERENCE COEFFICIENTS DEFINED IN TEXT
C ** A(I) THROUGH D(I) - FINITE DIFFERENCE COEFFICIENTS DEFINED IN TEXT
C ** BB(I),GG(I) - VARIABLES USED IN SOLUTION OF M SIMULTANEOUS EQUATIONS
C
DIMENSION T1(100)
DIMENSION T2(100)
DIMENSION U1(100)
DIMENSION U2(100)
DIMENSION CP(100)
DIMENSION CK(100)
DIMENSION X(100)
DIMENSION S2(100)
DIMENSION S3(100)
DIMENSION S4(100)
DIMENSION S5(100)
DIMENSION S6(100)
DIMENSION S7(100)
DIMENSION A(100)
DIMENSION B(100)
DIMENSION C(100)
DIMENSION D(100)
DIMENSION BB(100)
DIMENSION GG(100)
C
C*** DEFINE SYSTEM DIMENSIONS, MATERIAL PROPERTIES, AND APPROPRIATE CONSTANTS
W=1.0

```



```

Y=1.
Z=1.
RHO=1.80
SIG=1.356E-12
EM1=.85
TA=293.

C
C***DEFINE COUNTERS, SIZE FACTORS
K=2
M=100
M1=M-1
M2=M/2
DX=W/M
DT=1.E-6

C
C***SET INITIAL SPECIFIC ENERGY PROFILE
DATA U1/1800..1749..1697..1646..1594..1543..1491..1440..1389..1337
...1286..1234..1183..1131..1080..1029..977..926..874..823..771..
.720..669..617..566..514..463..411..360..309..257..206..154..
.103..51..13..64*0./

C
C***CALCULATE INITIAL TEMPERATURE PROFILE AND TOTAL ENERGY FROM SPECIFIC
C***ENERGY PROFILE
UTOT1=0.
DO 4 I=1,M
X(I)=DX*(I-1)
1 IF(U1(I).GT.228.5)GO TO 2
T1(I)=(4.0E3*U1(I)+TA**2)**.5
GO TO 3
2 T1(I)=2.*U1(I)+500.*(1.+1.0E-6*TA**2)
3 UTOT1=UTOT1+U1(I)*RHO*DX*Y*Z
4 CONTINUE
C
C***PRINT INITIAL VALUES
PRINT 5,W,Y,Z,RHO,EM1,TA,M,K
5 FORMAT(1H1,/,10X,*W= *.E10.4,* CM*,15X,*Y= *.E10.4,* CM*,14X,
1*Z= *.E10.4,* CM*,/,10X,*RHO= *.E10.4,* G/CM3*,10X,*EM1= *.E10.4,
215X,*TA= *.E10.4,* K*,/,10X,*M= *.I4,24X,*K= *.I4,/)
PRINT 6
6 FORMAT(10X,*LISTING OF INITIAL TEMPERATURE AND ENERGY PROFILES*,/)
PRINT 7
7 FORMAT(10X,2(1X,*I*,2X,*DISTANCE (CM)*,2X,*TEMPERATURE (K)*,2X,*ENER
*GY (CAL/G)*,2X))
DO 9 I=1,M2
I2=I+M2
PRINT 8,I,X(I),T1(I),U1(I),I2,X(I2),T1(I2),U1(I2)
8 FORMAT((10X,2(I3,3X,E10.3,3X,E11.4,6X,E10.4,3X)))
9 CONTINUE
C
C
C
C***CALCULATE VARIATION OF TEMPERATURE PROFILE AS A FUNCTION OF TIME
C
TIME=0.
DO 200 N=1,K
TIME=TIME+DT
C
C

```

```

C***CALCULATE THERMAL CONDUCTIVITY AND SPECIFIC HEAT FOR EACH NODE
  DO 12 I=1,M
    IF(T1(I).GT.1000.)GO TO 11
    CK(I)=.40-2.5E-4*T1(I)
    CP(I)=5.0E-4*T1(I)
    GO TO 12
  11  CK(I)=.1833-3.333E-5*T1(I)
     CP(I)=.50
  12  CONTINUE
C
C***CALCULATE VARIOUS FINITE DIFFERENCE COEFFICIENTS DEFINED IN TEXT
  S1=(Z+Y)*4.*(DX**2)/(Y*Z)
  S2(1)=4.*SIG*EM1*(T1(1)**3)
  S3(1)=(-3.*T1(1)**4-TA**4)*SIG*EM1
  S4(1)=DT/(2.*RHO*CP(1)*(DX**2))
  S5(1)=CK(2)+CK(1)
  A(1)=0.
  B(1)=-1.-S4(1)*S5(1)-S4(1)*(S1+2.*DX)*S2(1)
  C(1)=S4(1)*S5(1)
  D(1)=-T1(1)+S4(1)*(2.*DX+S1)*S3(1)
  DO 20 I=2,M1
    S2(I)=4.*SIG*EM1*(T1(I)**3)
    S3(I)=(-3.*T1(I)**4-TA**4)*SIG*EM1
    S4(I)=DT/(2.*RHO*CP(I)*(DX**2))
    S5(I)=CK(I+1)+CK(I)
    S6(I)=CK(I+1)+2.*CK(I)+CK(I-1)
    S7(I)=CK(I)+CK(I-1)
    A(I)=S7(I)*S4(I)
    B(I)=-1.-S4(I)*S6(I)-S1*S4(I)*S2(I)
    C(I)=S4(I)*S5(I)
    D(I)=-T1(I)+S4(I)*S1*S3(I)
  20  CONTINUE
     S2(M)=4.*SIG*EM1*(T1(M)**3)
     S3(M)=(-3.*T1(M)**4-TA**4)*SIG*EM1
     S4(M)=DT/(2.*RHO*CP(M)*(DX**2))
     S7(M)=CK(M)+CK(M-1)
     A(M)=S4(M)*S7(M)
     B(M)=-1.-S4(M)*S7(M)-S4(M)*(S1+2.*DX)*S2(M)
     C(M)=0.
     D(M)=-T1(M)+S4(M)*(2.*DX+S1)*S3(M)
C
C***PERFORM SOLUTION OF M SIMULTANEOUS EQUATIONS BY THOMAS ALGORITHM
  BB(1)=B(1)
  GG(1)=D(1)/BB(1)
  DO 30 I=2,M1
    BB(I)=B(I)-A(I)*C(I-1)/BB(I-1)
    GG(I)=(D(I)-A(I)*GG(I-1))/BB(I)
  30  CONTINUE
     BB(M)=B(M)-A(M)*C(M-1)/BB(M-1)
     GG(M)=(D(M)-A(M)*GG(M-1))/BB(M)
     T2(M)=GG(M)
     DO 40 I1=1,M1
       I=M-I1
       T2(I)=GG(I)-(C(I)*T2(I+1))/BB(I)
  40  CONTINUE
C
C***CALCULATE SPECIFIC ENERGY PROFILE AND TOTAL ENERGY AT NEW TIME

```

```

      UTOT2=0.
      DO 50 I=1,M
      IF(T2(I).GT.1000.)GO TO 42
      U2(I)=2.5E-4*(T2(I)**2-TA**2)
      GO TO 43
42     U2(I)=.5*T2(I)-250.*(1.+1.E-6*TA**2)
43     UTOT2=UTOT2+U2(I)*RHO*DX*Y*Z
50     CONTINUE
      ELOSS=(UTOT1-UTOT2)*100./UTOT1
C
C***PRINT OUTPUT FOR SELECTED TIMES
97     PRINT 98,TIME,N,DT
98     FORMAT(1H1,/,10X,*TIME=*,E10.3,* SEC*,5X,*N= *,15,5X,*DT= *,E10.4
      ..* SEC*,/)
      PRINT 99
99     FORMAT(10X,2(1X,*I*,2X,*DISTANCE(CM)*,2X,*TEMPERATURE(K)*,2X,*ENER
      *GY(CAL/G)*.2X))
      DO 101 I=1,M2
      I2=I+M2
      PRINT 100,I,X(I),T2(I),U2(I),I2,X(I2),T2(I2),U2(I2)
100    FORMAT((10X,2(I3,3X,E10.3,3X,E10.4,6X,E10.4,3X)))
101    CONTINUE
      PRINT 102,ELOSS
102    FORMAT(/,10X,*PERCENT ENERGY LOST= *,E10.4)
C
C***STOP PROGRAM WHEN T2(1) REACHES SPECIFIED LEVEL
105    TDIF=(T2(1)-TA)
      IF(TDIF.LT.50.)GO TO 300
C
C***SIZE DT (ADJUST MAGNITUDE OF TIME INCREMENT) SUCH THAT MAXIMUM TEMPERATURE
C***CHANGE WITHIN BLOCK IS 1 DEGREE K.
      AB=0.
      DO 106 I=1,M
      ABT=ABS(T2(I)-T1(I))
      IF(ABT.LT.AB)GO TO 106
      AB=ABT
106    CONTINUE
      DT=DT/AB
C
C***RESET TEMPERATURE PROFILE FOR NEW TIME STEP
109    DO 110 I=1,M
110    T1(I)=T2(I)
200    CONTINUE
C
C
300    CONTINUE
      STOP
      END

```

SECTION V

RESULTS

It was desired to investigate the thermal behavior of graphite calorimetry for the range of electron beam environments encountered in material and structural response testing. To accomplish this, the calorimeter system dimensions were fixed and the developed computer model was exercised for a variety of initial thermal loading conditions. The calorimeter block was taken to be square, 2 cm x 2 cm, and 1 cm in depth. These dimensions were later varied over a small range to study the effect of such variation.

As stated in the Introduction, the initial energy deposition profile is dependent upon the electron beam current and the voltage which accelerates the electrons. Frequently, however, the thermal loading conditions are described in terms of mean electron beam voltage and front surface specific energy (cal/g). For this study, linear energy deposition profiles were chosen to represent a range of electron beam environments. A total of twelve profiles were used; these profiles are shown in Figure 7.

Given the system dimensions and an initial energy deposition profile, the computer model provides calculation of the temperature profile history. An example of such a profile history is shown in Figure 8. The calculations are based upon a "1 MeV" electron beam with a front surface specific energy of 1200 cal/g. Through thermal conduction into the graphite block, the energy which was initially deposited in a non-uniform manner quickly







Figure 7. Electron Beam Deposition Profiles Used in This Study




Figure 8. Temperature Profiles for Calorimeter Block Subjected to a "1 MeV," 1200 cal/g Electron Beam

5
becomes more uniformly distributed. During this equilibration period, the front surface temperature is rapidly decreasing while that of the rear surface is rapidly increasing. Soon the energy becomes uniformly distributed, and the block then cools slowly through radiative losses to the surroundings.

For the example of Figure 8, the back surface temperature reaches a maximum value at a time of 3.07 seconds. This point in time corresponds to the peak in the cooling curve of Figure 2. As discussed in Section I, experimenters generally assume that the calorimeter block is at a uniform temperature when this maximum back surface value is attained. For the case presented in Figure 8, this is seen to be true. This trend was observed for all calculations performed under this study; at the time of maximum back surface temperature, the maximum and minimum temperatures within the block always differed by less than 0.5 percent. Thus, for the range of electron beam environments presented in Figure 7, the assumption of a uniform block temperature at the time of maximum back surface temperature is shown to be valid. This is considered a significant finding.

Shown in Figure 9 are continuous temperature histories for the back surface of a calorimeter block subjected to a "500 keV" beam. For front surface specific energies of 600, 1200, and 1800 cal/g, the peak back surface temperature occurs at 2.08, 2.19, and 2.56 seconds, respectively. Note that the temperature increases at the same rate for each case, but that the "cooling curve" becomes steeper as the front surface specific energy is increased. Thus, for a given mean electron energy, one would expect that as the front surface specific energy is increased, the total




Figure 9. Back Surface Temperature Histories for a "500 keV" Electron Beam

radiative losses at the time of maximum back surface temperature also increases.

In determining the total energy initially deposited in the calorimeter block, various extrapolations of the cooling curve are employed by experimenters. Given a back surface thermocouple measurement as shown in Figure 9, the most common correction involves a linear extrapolation of the cooling curve to time equal zero. The temperature obtained by this extrapolation is taken to be that which would result if the calorimeter block thermally equilibrated without any radiative losses to the surroundings. To obtain the total energy deposited in the block, this temperature is multiplied by the mass of the block and the specific heat of graphite at that temperature. The linear extrapolation is based upon the assumption that the calorimeter block cools at a constant rate. Much of the block, especially the front surface, actually radiates at a decreasing rate since the temperature is decreasing with time. For this reason, the linear extrapolation should underestimate the amount of initially deposited energy.

To quantify the error involved in such an extrapolation, the amount of energy actually lost through radiation at the time of maximum back surface temperature was investigated. These results are shown in Figure 10 for each of the environments of Figure 7. It is seen that the worst case for this energy loss is the "1 MeV," 1800 cal/g electron beam; at the time of peak back surface temperature, 9.5 percent of the initially deposited energy is lost.




Figure 10. Percent of Initially Deposited Energy Lost by Radiation at Time of Maximum Back Surface Temperature

Shown in Figure 11 is the back surface temperature history for the "1 MeV," 1800 cal/g electron beam. The amount of energy in the calorimeter block at the time of peak back surface temperature is given by that temperature to be 522 cal. Adjusting this value for the amount of energy lost through radiation, the energy initially deposited is calculated to be 577 cal. If a linear extrapolation of the cooling curve is employed, the temperature obtained yields an initially deposited energy of 546 cal. This amount is 5.4 percent below the correct value. Thus, although the linear extrapolation provides a partial correction for the radiative losses, it does indeed underestimate the initial energy in the block. Shown in Figure 11 is the extrapolation required to give the correct value for the initial energy.

It is seen that for the cases presented in Figure 10, the error involved in using the maximum back surface temperature to calculate the initially deposited energy is never more than ten percent. Furthermore, a linear extrapolation of the cooling curve reduces this error to a maximum of around five percent. Based upon these results, it is concluded that for a calorimeter of dimensions 2 cm x 2 cm x 1 cm, energy lost through radiation is not a significant problem, and a linear extrapolation of the cooling curve is an adequate means of accounting for such energy losses.

To investigate the effect of calorimeter dimensions on the amount of energy lost, the calorimeter area was varied from 1 cm² to 25 cm². It should be noted that to maintain conditions for one-dimensional heat conduction the size of the beam was always taken to be equal to the area of the calorimeter. Shown in Figure 12 are the results of this parametric




Figure 11. Back Surface Temperature History for Calorimeter Block Subjected to "1 MeV," 1800 cal/g Electron Beam




Figure 12. Effect of Calorimeter Area Upon Percent Energy Lost at Time of Maximum Back Surface Temperature

variation. It is seen that as the calorimeter area is increased above the 4 cm^2 used for the previous calculations, the effect of radiative losses becomes even less significant.

SECTION VI

CONCLUSIONS AND RECOMMENDATIONS

A computer model has been developed which provides calculation of the temperature profile and total energy within a graphite calorimeter block subjected to rapid energy deposition by an electron beam. The model is based upon an assumption of one-dimensional heat conduction with radiative energy transfer occurring at all free surfaces of the calorimeter block.

Through a series of parametric variations, the effectiveness of conventional graphite calorimetry has been investigated. Results indicate that for the range of electron beam environments typically employed in materials and structural response experiments, the problem of thermal radiation is not significant. For the worst case investigated, only ten percent of the initially deposited energy is lost before the calorimeter block reaches a uniform temperature. By means of a linear extrapolation of the cooling curve, the initially deposited energy can be computed to within five percent of the actual value.

The developed computer model can be used to further reduce the error involved in estimating the total deposited energy. Given information regarding the diode voltage and current histories, the peak back surface temperature measurement can be utilized to yield an initial energy deposition profile which does not provide any correction for radiative energy losses. Using this deposition profile in the developed computer code, the amount of energy lost through radiation at the time of maximum

back surface temperature can be calculated. Because the percent energy lost for the actual deposition profile and that obtained by the means described above will not differ considerably, the deposited energy given by the peak back surface temperature need only be adjusted by that amount.

These conclusions may contradict the actual experiences of some experimenters . It should be noted that the developed model assumes that the only mechanism for energy loss is through radiation to the surroundings. In real experimental configurations, there are always some paths for energy to leave the calorimeter through thermal conduction. An effort should be made towards more adequately thermally isolating the calorimeter system from its surroundings. If this could be done, then radiation would be the only energy loss mechanism. This study has shown that for such conditions, the problem of thermal radiation can be adequately accounted for.

LIST OF SYMBOLS

C	Specific heat, cal/g°K
E	Energy, cal
k	Thermal conductivity, cal/cm sec°K
M	Total number of differential elements
t	Time, sec
T	Temperature, °K
Q	Energy flux, cal/cm ² sec
W	Thickness of calorimeter block, cm
x	Location in calorimeter thickness, cm
Y	Height of calorimeter block, cm
Z	Width of calorimeter block, cm
Δx	Thickness of differential element, cm
Δt	Duration of calculational time step, sec
ϵ	Total hemispherical emissivity
σ	Stefan Boltzmann constant, cal/cm ² sec°K ⁴
ρ	Density, g/cm ³
β, γ	Factors for solution of simultaneous equations

Subscripts

i	Evaluated for i th differential element
n	Evaluated at n th time step
x	At the location $x=x$
$x+\Delta x$	At the location $x=x+\Delta x$
a	Evaluated at ambient conditions
o	Evaluated at some arbitrary value

Superscripts

k	Conduction
r	Radiation

DISTRIBUTION LIST

HEADQUARTERS USAF

AFTAC

ATTN: TAP

AFISC

ATTN: PQAL

MAJOR AIR COMMANDS

AUL

ATTN: LDE

AFIT

ATTN: Tech Library, Bldg 640, Area B

USAF

ATTN: SCLO (Maj J. H. Pierson,
Chief, LO)

AFSC ORGANIZATIONS

Air Force Materials Lab

ATTN: Tech Library

ATTN: Dr. Robert Craig

SAMSO

ATTN: Tech Library

ATTN: RSSE

ATTN: MNNR

AFATL

ATTN: DLOSL

AFRPL

ATTN: DYSN

KIRTLAND AFB ORGANIZATIONS

AFwL

ATTN: HO/Dr. Minge

2 cy ATTN: SUL

ATTN: DY

ATTN: DYV/Maj C. D. Stuber

ATTN: DYV/Maj R. F. Mitchell

ATTN: DYV/Mr. F. A. Bick

10 cy ATTN: DYV/Mr. K. D. Smith

ATTN: DYV/Mr. C. D. Newlander

ATTN: DYS/Dr. W. Baker

ATTN: ELS/Mr. B. Kline

ATTN: Lt John A. MacFarlane

ARMY ACTIVITIES

Army Materials and Mechanics Rsch. Center

ATTN: Mr. J. Dignam

ATTN: Dr. T. Chou

Harry Diamond Laboratories

ATTN: HDL Library

ATTN: Mr. D. Schallhorn

ATTN: Mr. P. Cladwell

ATTN: Mr. S. Graybill

ATTN: Mr. R. Oswald

ATTN: Mr. J. Gwaltney

Comdg Officer, BRL

ATTN: AMXBR-ED, Mr. H. Burden

Picatinny Arsenal

ATTN: Dr. P. Harris

NAVY ACTIVITIES

Director, Naval Research Laboratory

ATTN: Technical Library

2 cy ATTN: Dr. G. Cooperstein

Naval Surface Weapons Center

ATTN: Mr. L. Gowan

ATTN: Dr. J. Pastine

ATTN: Dr. J. Sazama (CASINO)

ATTN: Code 730, Technical Library

CO, NWEF

ATTN: ADS

OTHER DOD ACTIVITIES

Director, DNA

3 cy ATTN: STTL

2 cy ATTN: SPSS

ATTN: PPSR

ATTN: SPAS/Mr. J. Moulton

5 cy ATTN: SPAS/Mr. D. Kohler

5 cy ATTN: RAEV/Mr. J. Farber

ATTN: RAEV/Capt J. Van Prouyen

DDR&E

ATTN: Asst. Dir., Strat. Wpns.

Dir, DIA

ATTN: DI-7D

ATTN: DI-3

Dir, OSD, ARPA

ATTN: MMR

OTHER DOD ACTIVITIES (Continued)

Comdr, FC DNA
ATTN: FCTA

Chief, LVLO
ATTN: FCTC-DNA

JSTPS
ATTN: JLTW

DDC
12 cy ATTN: TC

ERDA ACTIVITIES

Sandia Laboratories
ATTN: Tech Library, Orgn 3141
ATTN: Dr. L. Posey
ATTN: Dr. B. Butcher
ATTN: Dr. A. Toepfer

Lawrence Livermore Laboratory
ATTN: Dr. J. Keller
ATTN: Mr. W. Isbell
ATTN: Dr. J. Moyer

Los Alamos Scientific Laboratory
ATTN: Report Library
ATTN: Dr. R. Skaggs
ATTN: Dr. R. Dingus

ADDITIONAL

Aeronutronic Ford Corp
ATTN: Mr. K. Attinger

Aerospace Corporation, ABRES Program
ATTN: Dr. B. Barry
ATTN: Dr. M. Kausch
ATTN: Dr. J. Benveniste

AVCO Corporation
3 cy ATTN: Dr. W. Bade
ATTN: Mr. W. Broding

The Boeing Aerospace Company
ATTN: Dr. B. Lempriere
ATTN: Mr. J. Adamski
ATTN: Mr. J. Shrader

Effects Technology
ATTN: Mr. B. Wengler
ATTN: Mr. M. Rosen

ADDITIONAL (CONTINUED)

General Electric Company
TEMPO-Center for Advanced Studies
ATTN: DASIAC

General Electric Co, Space Division
ATTN: Mr. G. Harrison
ATTN: Mr. J. Hannabeck
ATTN: Mr. R. G. Peterson

Ion Physics Corp
ATTN: Mr. B. Evans

Kaman Sciences Corp
ATTN: Dr. F. Shelton
ATTN: Mr. T. Meagher

KTech Corp
ATTN: Dr. D. V. Keller
ATTN: Mr. N. Froula
ATTN: Mr. L. Lee

Lockheed Missiles and Space Co, Inc,
ATTN: Dr. M. Miller
ATTN: Dr. A. O. Burford
ATTN: Mr. R. Smith, Dept 81-14
ATTN: Mr. R. Walz
ATTN: Mr. T. Kellcher

Maxwell Laboratories, Inc.,
ATTN: Dr. V. Fargo

McDonnell Douglas Astronautics Co.
ATTN: Dr. R. J. Reck
ATTN: Dr. H. Berkowitz

Physics International Co.
ATTN: Dr. J. Shea
ATTN: Mr. K. Childers
ATTN: Dr. S. Putnam

Prototype Development Associates, Inc.
ATTN: Mr. T. McKinley

Rand Corporation
ATTN: Mr. O. Nance

R&D Associates
ATTN: Dr. P. Rausch
ATTN: Dr. A. Field

Science Applications, Inc.
ATTN: Mr. R. Fisher

Science Applications, Inc.
ATTN: Dr. J. Cramer

ADDITIONAL (CONTINUED)

Simulation Physics, Inc.
ATTN: Mr. R. Little

Southern Research Institute
ATTN: Mr. Colt Pears

ADDITIONAL (CONTINUED)

Stanford Research Institute
ATTN: Mr. A. Lutze

Systems, Science and Software
ATTN: Dr. G. Gurtman

Official Record Copy (Mr. K. D. Smith, AFWL/DYV)